MAPLE Worksheet Number 12

Matrices and Linear Systems

One of the first places we usually encounter matrices is in studying linear systems. Suppose we want to solve the following system:

$$2\,x_1 + x_2 + 4\,x_3 = \frac{41}{4}$$

$$3\,x_1 + \frac{3\,x_2}{2} + \frac{34\,x_3}{5} = \frac{41}{12}$$

$$x_1 + \frac{2\,x_2}{3} + 5\,x_3 = \frac{35}{4}$$

One way would be to use the solve command on the three linear equations. Execute the following sequence.:

```
> L[1]:=2*x[1]+x[2]+4*x[3] = 41/4;
> L[2]:=3*x[1]+3/2*x[2]+34/5*x[3] = 41/12;
> L[3]:=x[1]+2/3*x[2]+5*x[3] = 35/4;
> solve({L[1],L[2],L[3]},{x[1],x[2],x[3]});
```

Another approach to solving the system is to convert it into matrix form and perform matrix operations. Execute the following sequence.

```
> with(linalg):
> L:=matrix([[coeffs(lhs(L[1]))],[coeffs(lhs(L[2]))],[coeffs(lhs(L[3
  ]))]]);
```

This gives the matrix associated with the lhs of the linear system. Define the column vector V to be

```
> V:=matrix(3,1,[x[1],x[2],x[3]]);
> L&*V;
> evalm(%);
```

Define the column matrix

```
> B:=matrix(3,1,[41/4,41/12,35/4]);
```

So the matrix form of our linear system is

```
> L&*V=B;
> Meq:=evalm(%);
> L^(-1)&*lhs(Meq)=L^(-1)&*rhs(Meq);
> evalm(%);
```

Of course this is the same solution obtained above. A third approach the solving the original system involves manipulating the "augmented matrix" for the original system. This involves relegating everything to a matrix manipulation devoid of any variable expressions. The augmented matrix for the system is

```
> ALB:=augment(L,B);
```

Notice the first three columns give the coefficients of x_1, x_2, and x_3 respectively. The idea is to preform "row operations" which reduce the matrix to one repesenting a system for which the solution is obvious. This technique is based on the principle that "equal actions on equal objects produce equal results." Preform the following command sequence, after each command say, on the worksheet, what each command did to the previous matrix, what actions it corresponds to on the corresponding system and why that action is valid (meaning it doesn't affect the solution).

```
> ALB1:=swaprow(ALB,1,3);
> mulrow(ALB1,2,1/3);
> ALB2:=mulrow(%,3,1/2);
> addrow(%,1,2,-1);
> ALB3:=addrow(%,1,3,-1);
> addrow(ALB3,2,3,-1);
> mulrow(%,2,-6);
> ALB4:=mulrow(%,3,-15/4);
```

Use the submatrix command to obtain the linear system corresponding to ALB4.

```
> submatrix(ALB4,1..3,1..3)&*V=submatrix(ALB4,1..3,4..4);
> evalm(%);
```

The solution is obvious by "backward substitution" beginning with x_3 , in MAPLE the command is

```
> soln:=backsub(ALB4);
```

The technique to arrive at ABL4 is called "Gaussean Elimination" and could have been sort-of achieved immediately in this example with the command

```
> gausselim(ALB);
```

We say "sort-of" because this row echelon form is not unique, as we see from the above two examples. However the solution obtained is the same in either case.

```
> backsub(%);
```

Continued use of row operations on this matrix could produce the uniquely determined "row reduced echelon form" for ALB4. There are two commands for producing this form of a matrix directly from ALB, the easiest to remember is

```
> rref(ALB);
```

The other is

```
> gaussjord(ALB);
```

This corresponds to a linear system that even a Texas A&M student could solve. What is it?